************************************************************************************************************************
*This is a replication do-file for "Antidote to Backsliding: Ethnic Politics and Democratic Resilience" published in the APSR, 
*author: Jan Rovny, jan.rovny@sciencespo.fr 
*November 13, 2022 
*compiled with Stata MP 16.1 
*requires "eststo", "plottig"  ado files 
************************************************************************************************************************



********************************************************************************************
*1) Graphing backsliding -- Figure 1 
********************************************************************************************
use "Backsliding_data.dta", clear 

*Backsliding as accumulation of 'back steps' 
tsset cid year 
gen back=d.v2x_libdem
replace back=0 if back>0
gen cumback=-1*back // make backsliding measure increase 
collapse (sum) cumback (mean) vote_eth, by(cid) // back is a cummulative measure of all "back steps" in liberal democracy

*generate region of EEu
gen re=.
replace re=0 if cid==3 | cid==5 | cid==8  // CZ, HU, PL, 
replace re=1 if cid==1 | cid==2 | cid==4 | cid==6 | cid==7 | cid==9 | cid==10 | cid==11  // BG, HR, EE, LV, LT, RO, SK, SI
label de re 0 "homogenous" 1 "heterogenous"
label val re re 

bys re: sum cumback

twoway (scatter cumback vote_eth if cid!=10, mlabel(cid) mlabpos(0) msymbol(i)) ///
(scatter cumback vote_eth if cid==10, mlabel(cid) mlabpos(6) msymbol(i) mlabcolor(black)) ///
(lfit cumback vote_eth, lcolor(midblue)) ///
, scheme(plottig) yti(Cumulative backsliding) xti(Ethnic party vote) xsc(r(-2 22)) legend(off) saving(backsliding.gph, replace)
graph export Fig1.jpg, replace 


********************************************************************************************
*2) Graphing liberal democracy over time -- Figure 2 
********************************************************************************************
use "Backsliding_data.dta", clear 

*graph dependent variable 
twoway line v2x_libdem year, scheme(plottig) yti(Liberal democracy) xti(Year) by(cid, note(""))
graph export Fig2.jpg, replace

********************************************************************************************
*3) Testing vote for constitutional liberal parties -- Figure 3 
********************************************************************************************
use "Backsliding_data.dta", clear 

*generate region of EEu
gen re=.
replace re=0 if cid==3 | cid==5 | cid==8  // CZ, HU, PL, 
replace re=1 if cid==1 | cid==2 | cid==4 | cid==6 | cid==7 | cid==9 | cid==10 | cid==11  // BG, HR, EE, LV, LT, RO, SK, SI
label de re 0 "homogenous" 1 "heterogenous"
label val re re 

ttest vote_lib, by(re) 

collapse (mean) meanvote= vote_lib (sd) sdvote=vote_lib (count) n=vote_lib, by(re)

*error bars
gen hivote=meanvote+invttail(n-1,0.025)*(sdvote/sqrt(n))
gen lovote=meanvote-invttail(n-1,0.025)*(sdvote/sqrt(n))

*graph  
graph twoway (bar meanvote re, color(gs5)) (rcap hivote lovote re, color(black)), scheme(plottig) legend(off) ///
	ti("Vote for constitutional liberal parties") yti("mean vote share") xlab(0 "no" 1"yes") xti("mobilized ethnic minorities") ///
	note("difference: p<0.01") saving(lib_vote_reg.gph, replace) ysc(r(0 25)) ylab(0(5)25)

	graph export Fig3.jpg, replace 


********************************************************************************************
*4) Time-series cross-section analyses (table 1, figures 4 and 5)
********************************************************************************************
use "Backsliding_data.dta", clear 


eststo clear
tsset cid year

*******************************************		
*Model 1, table 1: Ethnic vote
eststo M1: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)
	margins , dydx(vote_rr) atmeans at(vote_eth=(0 5 12) col_gov=0 eu_mem=1) post  // get the partial slopes at 3 levels of vote_eth
		test _b[1._at] = _b[2._at] // test slope differences 
		test _b[1._at] = _b[3._at]
		test _b[2._at] = _b[3._at]

eststo M1: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)		
	margins , atmeans at(vote_rr=(0(5)70) vote_eth=(0 5 12) col_gov=0 eu_mem=1) 	
	marginsplot, recast(line) recastci(rarea) ///
		plot1opts(fintensity(50) lpattern(solid)) ci1opts(fintensity(20) fcolor(black) lcolor(gs12) ) /// 
		plot2opts(fintensity(50) lpattern(shortdash) lcolor(blue)) ci2opts(fintensity(20) fcolor(black) lcolor(gs12)) ///
		plot3opts(fintensity(50) lpattern(shortdash_dot) lcolor(red)) ci3opts(fintensity(20) fcolor(black) lcolor(gs12)) ///
		scheme(plottig) ti("Ethnic effect") legend(order(4 "0%" 5 "5%" 6 "12%") subtitle("ethnic vote", size(small)) position(6) row(1)) ///
		xti("illiberal vote (%)") ylab(0.4(0.1)0.8) xlab(0(10)70) yti("predicted democracy") saving(rr.gph, replace) ///
		text(0.75 65 "{&beta}= 0.0005", color(red) size(small)) ///
		text(0.66 65 "{&beta}= -0.0016*", color(blue) size(small)) ///
		text(0.52 65 "{&beta}= -0.0032*", color(black) size(small))

**************
*Figure 5***
	margins , atmeans at(col_gov=(0 1) eu_mem=0) post 
	 test _b[1._at] = _b[2._at] // test difference 
	marginsplot, recast(scatter) scheme(plottig) yti("predicted democracy") xti("ethnic party government participation") xlab(0 "no" 1"yes") ti("") xsc(r(-0.2 1.2)) saving(col_gov.gph, replace)
	graph export Fig5.jpg, replace // figure 5 
		
	
*******************************************	
*Model 2: Liberal vote 
eststo M2: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)
	margins , dydx(vote_rr) atmeans at(vote_lib=(0 20 35) col_gov=0 eu_mem=1) post  // get the partial slopes at 3 levels of vote_lib
		test _b[1._at] = _b[2._at] // test slope differences 
		test _b[1._at] = _b[3._at]
		test _b[2._at] = _b[3._at]
		
eststo M2: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)
		
	margins , atmeans at(vote_rr=(0(5)70) vote_lib=(0 20 35) col_gov=0 eu_mem=1) 
	marginsplot, recast(line) recastci(rarea) ///
		plot1opts(fintensity(50) lpattern(solid)) ci1opts(fintensity(20) fcolor(black) lcolor(gs12) ) /// 
		plot2opts(fintensity(50) lpattern(shortdash) lcolor(blue)) ci2opts(fintensity(20) fcolor(black) lcolor(gs12)) ///
		plot3opts(fintensity(50) lpattern(shortdash_dot) lcolor(red)) ci3opts(fintensity(20) fcolor(black) lcolor(gs12)) ///
		scheme(plottig) ti("Const. liberal effect") ylab(0.4(0.1)0.8) xlab(0(10)70) legend(order(4 "0%" 5 "20%" 6 "35%") subtitle("const. liberal vote", size(small)) position(6) row(1)) ///
		xti("illiberal vote (%)") yti("") saving(li.gph, replace) ///
		text(0.68 65 "{&beta}= -0.0013*", color(red) size(small)) ///
		text(0.59 65 "{&beta}= -0.0022*", color(blue) size(small)) ///
		text(0.51 65 "{&beta}= -0.0034*", color(black) size(small)) 
	
*******************************************	
*Model 3: Combined ethnic and liberal vote 
eststo M3: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)	
	margins , dydx(vote_rr) atmeans at(vote_comb=(2.5 25 45) col_gov=0 eu_mem=1) post  // get the partial slopes at 3 levels of vote_comb
		test _b[1._at] = _b[2._at] // test slope differences 
		test _b[1._at] = _b[3._at]
		test _b[2._at] = _b[3._at]
		
eststo M3: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)			
	margins , atmeans at(vote_rr=(0(5)70) vote_comb=(2.5 25 45) col_gov=0 eu_mem=1) 
	marginsplot, recast(line) recastci(rarea) ///
		plot1opts(fintensity(50) lpattern(solid)) ci1opts(fintensity(20) fcolor(black) lcolor(gs12) ) /// 
		plot2opts(fintensity(50) lpattern(shortdash) lcolor(blue)) ci2opts(fintensity(20) fcolor(black) lcolor(gs12)) ///
		plot3opts(fintensity(50) lpattern(shortdash_dot) lcolor(red)) ci3opts(fintensity(20) fcolor(black) lcolor(gs12)) ///
		scheme(plottig) ti("Combined effect") ylab(0.4(0.1)0.8) xlab(0(10)70) legend(order(4 "2.5%" 5 "25%" 6 "45%") subtitle("combined ethnic & const. liberal vote", size(small)) position(6) row(1)) ///
		xti("illiberal vote (%)") yti("") saving(co.gph, replace) ///
		text(0.72 65 "{&beta}= -0.0004", color(red) size(small)) ///
		text(0.60 65 "{&beta}= -0.0020*", color(blue) size(small)) ///
		text(0.50 65 "{&beta}= -0.0037*", color(black) size(small)) 


	
*prepare table 1  			
esttab M1 M2 M3 using Table1.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted nobase ///
			 order(vote_rr vote_eth c.vote_rr#c.vote_eth vote_lib c.vote_rr#c.vote_lib vote_comb c.vote_rr#c.vote_comb col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) /// 
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 

*******************************************	
*Figure 4 
graph combine rr.gph li.gph co.gph , scheme(plottig) row(1) col(3)
graph export Fig4.jpg, replace 
	

********************************************************************************************
*5) Czech and Slovak case study (figures 6 and 7)
********************************************************************************************
use "Backsliding_data.dta", clear 
keep if cid==3 | cid==10

*get lib dem values at specific years 
sum v2x_libdem if cid==3 // CZ max=.833 
sum v2x_libdem if cid==10 // SK max=.767

sum v2x_libdem if cid==10 & year==1993 // before Meciar = .64
sum v2x_libdem if cid==10 & year==1993 // at Meciar = .544
sum v2x_libdem if cid==3 & year==2012 // before Babis = .821
sum v2x_libdem if cid==10 & year==2011 // before Fico = .767
sum v2x_libdem if cid==3 & year==2020 // CZ 2020=.708
sum v2x_libdem if cid==10 & year==2018 // SK 2018=.703


*****************
* Liberal democracy in CZ and SK -- Figure 6 
twoway (line v2x_libdem year if cid==3, lpattern(solid)) (line v2x_libdem year if cid==10, lpattern(dash)) ///
	(pcarrowi .821 2020.5 .708 2020.5 (3) "backsliding Babiš (CZ)", mcolor(black) lcolor(black) mlabcolor(black)) ///
	(pcarrowi .767 2020.8 .703 2020.8 (3) "backsliding Fico (SK)", mcolor(midblue) lcolor(midblue) mlabcolor(midblue)) ///
	(pcarrowi .64 2020.5 .544 2020.5 (3) "backsliding Mečiar (SK)", mcolor(midblue) lcolor(midblue) mlabcolor(midblue)), ///
	text(0.53 1995 "Mečiar", place(e) color(midblue) size(vsmall)) ///
	text(0.77 1998 "Dzurinda", place(e) color(midblue) size(vsmall)) ///
	text(0.765 2011.9 "Fico", place(e) color(midblue) size(vsmall)) ///
	text(0.82 2012.5 "Babiš", place(e) color(black) size(vsmall)) ///
	yti(liberal democracy) xti(year) scheme(plottig) legend(order(1 "Czech Republic" 2 "Slovakia"))
	graph export Fig6.jpg, replace

***************************	
*Liberal vote in CZ and SK -- Figure 7 
twoway (line vote_lib year if cid==3, lpattern(line)) ///
	(line vote_lib year if cid==10 & year>1992, lpattern(dash)) ///
	(line vote_comb year if cid==10 & year>1992, lpattern(shortdash_dot) lcolor(red)), scheme(plottig) yti("vote share") legend(order(1 "Czech const. liberals" 2 "Slovak const. liberals" 3 "Slovak const. liberal & ethnic parties") pos(6) row(1))
	graph export Fig7.jpg, replace
	
********************************************************************************************	
********************************************************************************************
* APPENDIX ANALYSES 
********************************************************************************************
********************************************************************************************


********************************************************************************************
*Figure A1 - Comparign V-Dem, Polity and Freedom House Measures in Hungary 
use "Backsliding_data.dta", clear 	
	
*recode all measures of democracy to be 0-1 and increasing democracy
gen pol=(polity2+10)/20
gen f_c=-((fh_cl-1)/6)+1
gen f_p=-((fh_pr-1)/6)+1

*graph
twoway  (line v2x_libdem year, lpattern(solid)) ///
	(line pol year, lpattern(dash)) ///
	(line f_c year, lpattern(-.) lcolor(red)) ///
	(line f_p year, lpattern(--..) lcolor(orange)) if cid==5, ///
	scheme(plottig) legend( label(1 "V-Dem Liberal Democracy") label(2 "Polity") label(3 "FH Civil Liberties") label(4 "FH Political Rights"))
graph export dem_measures.jpg, replace 


********************************************************************************************
*Table A2 - factor analysis of V-Dem measures of democracy
use "Backsliding_data.dta", clear 	


factor v2x_polyarchy v2x_libdem v2x_partipdem v2x_delibdem v2x_egaldem v2x_api v2x_mpi
predict dem_pf


********************************************************************************************
*Figure A2 - Party types and anti-pluralism 
use "Party_type.dta", clear

reg v2xpa_antiplural i.ptype 
margins ptype
marginsplot, recast(scatter) xlab(4 "other" 1"ethnic" 2"illiberal" 3"constitutional liberal") scheme(plottig) xti(party type) yti(Anti-pluralism) ti("") ysc(r(0 0.6)) ylab(0(0.2)0.6)
graph export Fig_A2.jpg, replace


********************************************************************************************
*Table A8 Alternative Party Classification 	

use "Backsliding_data_alternative.dta", clear 	

eststo clear
tsset cid year

*Ethnic parties:
eststo M1: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

	
*Liberal parties 
eststo M2: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*Combined 	
eststo M3: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)
			
			
esttab M1 M2 M3 using Tab_A8.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted nobase ///
			 order(vote_rr vote_eth c.vote_rr#c.vote_eth vote_lib c.vote_rr#c.vote_lib vote_comb c.vote_rr#c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) /// 
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 
	
********************************************************************************************
*Figure A3 - Robustness of key beta coefficients to various specification of illiberal and constitutional liberal party categories 
 
 **use "Randomization_analyses.do"


********************************************************************************************
*Table A9 - Robustness checks with alternative models - ethnic party vote share  

use "Backsliding_data.dta", clear 

eststo clear
tsset cid year
	

*create democracy factor
factor v2x_polyarchy v2x_libdem v2x_partipdem v2x_delibdem v2x_egaldem v2x_api v2x_mpi
predict dem_pf

*ML model 
eststo CE1: xtmixed v2x_libdem (c.vote_rr)##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem || cid:

*alternative DV 
eststo CE2: xtreg dem_pf c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*LDV model
eststo CE3: xtreg v2x_libdem l.v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*Country and Year FE model
eststo CE4: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov i.year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

***Bootstrap SEs 
eststo BS1: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid) vce(bootstrap , rep(10) seed(321))	

esttab CE1 CE2 CE3 CE4 BS1 using Tab_A9.tex, replace   ///
	order(vote_rr vote_eth c.vote_rr#c.vote_eth col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted ///  
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 

********************************************************************************************
*Table A10 - Robustness checks with alternative models - constitutional liberal party vote share  

*ML model
eststo CL1: xtmixed v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem || cid:

*alternative DV 
eststo CL2: xtreg dem_pf c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*LDV model
eststo CL3: xtreg v2x_libdem l.v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*Country and Year FE model
eststo CL4: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov i.year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*Bootstrap SE 
eststo BS2: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid) vce(bootstrap , rep(10) seed(321))	

esttab CL1 CL2 CL3 CL4 BS2 using Tab_A10.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted /// 
			order(vote_rr vote_lib c.vote_rr#c.vote_lib col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) ///
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 
	
********************************************************************************************
*Table A11 - Robustness checks with alternative models - comb eth and const lib vote share  
	
*ML model
eststo CC1: xtmixed v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem || cid:

*alternative DV 
eststo CC2: xtreg dem_pf c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*LDV model
eststo CC3: xtreg v2x_libdem l.v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*Country and Year FE model
eststo CC4: xtreg v2x_libdem cc.vote_rr##c.vote_comb col_gov i.year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

*Bootstrap SEs
eststo BS3: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid) vce(bootstrap , rep(10) seed(321))

esttab CC1 CC2 CC3 CC4 BS3 using Tab_A11.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted /// 
			order(vote_rr vote_comb c.vote_rr#c.vote_comb col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) ///
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 	

********************************************************************************************
*Table A12 - Models without extreme cases 	

*without EE and LV 
eststo W1: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem if cid!=4 & cid!=6, fe cl(cid)
eststo W2: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem if cid!=4 & cid!=6, fe cl(cid)
eststo W3: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem if cid!=4 & cid!=6, fe cl(cid)

*Without HU and PL 
eststo W4: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem if cid!=5 & cid!=8, fe cl(cid)
eststo W5: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem if cid!=5 & cid!=8, fe cl(cid) 
eststo W6: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem if cid!=5 & cid!=8, fe cl(cid) // 


esttab W1 W2 W3 W4 W5 W6 using Tab_A12.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted nobase ///
			 order(vote_rr vote_eth c.vote_rr#c.vote_eth vote_lib c.vote_rr#c.vote_lib vote_comb c.vote_rr#c.vote_comb col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) /// 
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 	
	
********************************************************************************************
*Table A13 - Models controlling for anti-pluralism of other parties 

eststo Q1: xtreg v2x_libdem c.vote_rr##c.vote_eth col_gov year anti nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)
eststo Q2: xtreg v2x_libdem c.vote_rr##c.vote_lib col_gov year anti nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)
eststo Q3: xtreg v2x_libdem c.vote_rr##c.vote_comb col_gov year anti nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem, fe cl(cid)

esttab Q1 Q2 Q3 using Tab_A13.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted nobase ///
			 order(vote_rr vote_eth c.vote_rr#c.vote_eth vote_lib c.vote_rr#c.vote_lib vote_comb c.vote_rr#c.vote_comb col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) /// 
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 	

********************************************************************************************
*Table A14 - Alternative models with LDV and year FE

eststo Z1: xtreg v2x_libdem l.v2x_libdem c.vote_rr##c.vote_eth col_gov  nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem i.year, fe cl(cid)  
eststo Z2: xtreg v2x_libdem l.v2x_libdem c.vote_rr##c.vote_lib col_gov  nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem i.year, fe cl(cid) 
eststo Z3: xtreg v2x_libdem l.v2x_libdem c.vote_rr##c.vote_comb col_gov  nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem i.year, fe cl(cid)	

esttab Z1 Z2 Z3 using Tab_A14.tex, replace   ///
			b(4) se(3) ar(3) sca(sigma_u sigma_e) staraux label noomitted nobase ///
			 order(vote_rr vote_eth c.vote_rr#c.vote_eth vote_lib c.vote_rr#c.vote_lib vote_comb c.vote_rr#c.vote_comb col_gov ///
			 year nygdppcapppkd_i gini_disp_i wdi_unempilo_i icrg_qog_i eu_mem) /// 
	star(+ 0.10 * 0.05 ** 0.01 *** 0.001)  ///
	eqlabels(none) 
